SQUELCH THOSE WHIPSAWS

John Ehlers

Hey good buddy!  Ever operate a CB radio?  Most CBs only have three controls; a channel selector, a volume control, and a squelch control.  Since there are only three controls, and since squelch is one of them, it must be important to the operation of the radio.  Such a control can also be important to your trading.


What is a squelch control and how does it work?  If you turn the squelch to maximum, all you get out of the CB radio is noise.  Turning it down, the noise goes away, and is “squelched”.  Then, when a voice comes on you hear a short burst of noise just before you hear the voice.  The principle of operation is that the radio receives a range of frequencies in each channel.  The fundamentals of the voice waves happen to be located at the lower portion of the frequency spectrum.  On the other hand, noise is spread across the entire spectrum.  By filtering the low frequencies, the amplitude of signals in the lower part of the spectrum can be compared to the amplitude of signals in the higher part of the spectrum.  If noise only is present, the power in the two parts of the spectrum will be roughly equal.  When voice is present, the amplitude of the lower part of the spectrum will be larger.   The squelch control makes an amplitude comparison of the two parts of the spectrum.  If the amplitude in the higher frequency part of the spectrum is about equal to the amplitude in the lower part of the spectrum, then signals to the speakers is switched off by the squelch control.  When a voice comes over the channel, the lower frequency part of the channel has higher amplitude and the signals are switched through to the speakers by the squelch control.  This keeps you from being bothered by noise during all the time there is no voice being received.


If you are a trend mode trader, you are bothered by the consolidation phase of the market (I prefer to call this the cycle mode of the market).  From a signal analysis perspective, the difference between these two modes is the length of the cyclic period.  A trend can be viewed as a piece of a long cycle.  A long cycle period means the frequency of the cycle is low.   Cycle mode signals tend to have shorter cycle periods.  With this distinction between market modes, we can design an indicator to flag trend modes just like a squelch control on a radio switches voice through to the speakers.


It turns out that designing precision low frequency filters in conventional trading software is difficult because the rounding errors in the calculations cause the filters to be unstable.  (This fact alone makes this article worth while, saving you the many hours I spent trying to design stable precision filters.)  Therefore, it is impractical to design a squelch circuit in exactly the same way it is done in a CB radio by comparison of filter outputs.  However, I described a way to measure the dominant cycle.
  If the dominant cycle period is long we can assume that most of the energy is in the low frequency part of the spectrum.  If the dominant cycle period is shorter, we can assume that most of the energy is in the higher frequency part of the spectrum.  So, to implement a squelch circuit for trading, all we have to do is to compare the measured dominant cycle to some fixed threshold at which the squelch is set.  If the measured dominant cycle period is longer than the threshold value, the market is in a trend mode.  If the measured dominant cycle period is shorter than the threshold value, the market is in a cycle mode.  The threshold value can be almost anything, but a good starting point is 20 bars, corresponding to a one month cycle using daily data.


The EasyLanguage Code to implement the Squelch Indicator is shown in the sidebar.  This code is identical to the Hilbert Period code given in Reference 1, with the addition of the squelch threshold and the display being implemented as a paintbar.   With reference to Figure 1, when the measured dominant cycle (shown in the subgraph) is shorter than the 20 bar squelch threshold, the market is in a cycle mode and the bars are painted red.  Otherwise, the measured dominant cycle is longer than the squelch threshold and the Trend Mode is identified by the green price bars.  The Trend Mode can be identified sooner, and extend longer, when the Squelch threshold is turned down to 15 bars, as shown in Figure 2.  Lowering the Squelch Threshold still further can result in identifying virtually all the bars as being in the Trend Mode.  This is just about the same as setting the squelch on the radio to hear noise all the time.


So there you have it good buddy – a simple Squelch control that can keep you out of the market during those times that whipsaws are most likely.  Over and out. 

Figure 1.  A Squelch setting of 20 clearly identifies two profitable trends.

Figure 2.  A Squelch setting of 16 gives earlier identification of the beginning and extends the duration of the Trend Modes, but with additional uncertainty.

****************************************************************************************************

SIDEBAR


EasyLanguage code to implement the Squelch Control as a Paintbar

Inputs:
Price((H+L)/2),



Squelch(20);

Vars:

InPhase(0),



Quadrature(0),



Phase(0),



DeltaPhase(0),



count(0),



InstPeriod(0),



Period(0);

If CurrentBar > 5 then begin


{Compute InPhase and Quadrature components}


Value1 = Price - Price[6];


Value2 =Value1[3];


Value3 =.75*(Value1 - Value1[6]) + .25*(Value1[2] - Value1[4]);


InPhase = .33*Value2 + .67*InPhase[1];


Quadrature = .2*Value3 + .8*Quadrature[1];


{Use ArcTangent to compute the current phase}


If AbsValue(InPhase +InPhase[1]) > 0 then Phase = ArcTangent(AbsValue((Quadrature+Quadrature[1]) / (InPhase+InPhase[1])));


{Resolve the ArcTangent ambiguity}


If InPhase < 0 and Quadrature > 0 then Phase = 180 - Phase;


If InPhase < 0 and Quadrature < 0 then Phase = 180 + Phase;


If InPhase > 0 and Quadrature < 0 then Phase = 360 - Phase;


{Compute a differential phase, resolve phase wraparound, and limit delta phase errors}


DeltaPhase = Phase[1] - Phase;


If Phase[1] < 90 and Phase > 270 then DeltaPhase = 360 + Phase[1] - Phase;


If DeltaPhase < 1 then DeltaPhase = 1;


If DeltaPhase > 60  then Deltaphase = 60;


{Sum DeltaPhases to reach 360 degrees.  The sum is the instantaneous period.}


InstPeriod = 0;


Value4 = 0;


For count = 0 to 40 begin



Value4 = Value4 + DeltaPhase[count];



If Value4 > 360 and InstPeriod = 0 then begin




InstPeriod = count;



end;


end;


{Resolve Instantaneous Period errors and smooth}


If InstPeriod = 0 then InstPeriod = InstPeriod[1];


Period = .25*InstPeriod + .75*Period[1];


If Period < Squelch then begin 



Plot1(High,"High");



Plot2(Low,"Low");


end;

end;

